CENTRE of POLICY STUDIES and the IMPACT PROJECT

Eleventh Floor
Menzies Building
Monash University Wellington Road
CLAYTON Vic 3168 AUSTRALIA

Telephone: (03) 565 2398, (03) 565 5112
from overseas: 61 3 565 2398 or 61 3 565 5112

Fax numbers: (03) 565 2426, (03)565 5486
from overseas: 61 3 565 2426 or 61 3 565 5486

e-mail impact@vaxc.cc.monash.edu.au

Companion Paper to "Solving Applied General Equilibrium Models represented as a Mixture of Linearized and Levels Equations" by W. Jill Harrison, K.R. Pearson, Alan A. Powell and E. John Small, Centre of Policy Studies and Impact Project Preliminary Working Paper No. IP-61, September 1993

TABLO INPUT FILES FOR THE STYLIZED JOHANSEN, MINIATURE ORANI AND ORANI-F MODELS

by

W. Jill HARRISON
Monash University
and
E. John SMALL
Australian Bureau of Agricultural and Resource Economics

Computing Document No. C11-01 September 1993

ISSN 1 031 9034 ISBN 0 642 10305 4

Abstract and Explanatory Note

This document is a companion paper to "Solving Applied General Equilibrium Models Represented as a Mixture of Linearized and Levels Equations" by Harrison, Pearson, Powell and Small (Centre of Policy Studies and Impact Project *Preliminary Working Paper* No. IP-61, September 1993). The latter paper describes the new *GEMPACK* facility for solving models which contain both levels and linearized equations, and discusses the advantages of using a mixed representation. The discussion in IP-61 is illustrated by reference to three models:

- Stylized Johansen(STJ) (a tiny teaching model)
- Miniature ORANI (MO) (a somewhat larger miniature model)
- ORANI-F (the standard forecasting version of ORANI).

This document contains the complete TABLO Input files for the mixed representations of these three models described in sections 2, 3 and 5 of IP-61. References to the literature will be found in IP-61.

If you have access to the *GEMPACK* software and wish to run any of these three models, a companion disk containing the TABLO Input files shown below, plus several auxiliary files used in running simulations on a 80386/80486 PC, is available on request from the authors (Please send a blank floppy with your request).

Contents

```
!-------------------------------------------------------------------!
!                  Mixed TABLO Input file for the                   !
!                    Stylized Johansen model                        !
!                                                                   !
!         following the description in Chapter 3 of the text        !
!   "Notes and Problems in Applied General Equilibrium Economics"   !
!       by P.Dixon, B.Parmenter, A.Powell and P.Wilcoxen [DPPW]     !
!              published by North-Holland 1992.                     !
!                                                                   !
!    This version is as described in Working Paper No. IP-61        !
!    "Solving Applied General Equilibrium Models represented        !
!      as a Mixture of Linearized and Levels Equations "            !
!      by W.Jill Harrison, K.R.Pearson, Alan.A.Powell and           !
!      E.John Small - [HPPS]   September 1993.                      !
!-------------------------------------------------------------------!
 ! This TABLO Input file is an implementation of the shaded equations
   in Table 2.1 of the paper HPPS !

VARIABLE (DEFAULT = LEVELS) ;
FORMULA (DEFAULT = INITIAL) ;
COEFFICIENT (DEFAULT = PARAMETER) ;

 ! The index set { i : i = 1,2,3,4 } in Table 2.1 of the paper HPPS
   corresponds to the union of the sets SECT and FAC below. The set
   SECT doubles for commodities {i:i = 1,2} and industries {j:j=1,2}. !

SET SECT # Sectors # (i1-i2)  ;

SET FAC # Factors - labour and capital # (i3-i4) ;

VARIABLE
                Y # Total nominal household expenditure # ;
(all,i,SECT)    XH(i)   # Household demand for commodity i # ;
(all,i,SECT)    XCOM(i) # Total demand for commodity i # ;
(all,f,FAC)     XFAC(f) # Total demand for factor f    # ;
(all,i,SECT)    PC(i)   # Price of commodity i # ;
(all,f,FAC)     PF(f)   # Price of factor f #;
 (all,i,SECT)(all,j,SECT)     XC(i,j)
    # Intermediate inputs of commmodity i to industry j # ;
(all,f,FAC)(all,j,SECT)      XF(f,j) # Factor inputs to industry j # ;

(all,i,SECT)(all,j,SECT)      DVCOMIN(i,j)
 # Dollar value of inputs of commodity i to industry j # ;
(all,f,FAC)(all,j,SECT)      DVFACIN(f,j)
 # Dollar value of factor f used in industry j # ;
(all,i,SECT)                  DVHOUS(i)
 # Dollar value of household use of commodity i # ;

COEFFICIENT
(all,i,SECT)(all,j,SECT)  alphacom(i,j) ;
(all,i,FAC)(all,j,SECT)   alphafac(i,j) ;

FILE iodata # input-output data for the model # ;

READ DVCOMIN from FILE iodata HEADER "CINP" ;
READ DVFACIN from FILE iodata HEADER "FINP" ;
READ DVHOUS  from FILE iodata HEADER "HCON" ;

FORMULA (all,i,SECT) PC(i) = 1.0 ;
FORMULA (all,i,FAC) PF(i) = 1.0 ;

FORMULA
   ! Initial Share of intermediate commodity i in costs of industry j !
(all,i,SECT)(all,j,SECT) alphacom(i,j) = DVCOMIN(i,j) /
          (SUM(ii,SECT,DVCOMIN(ii,j)) + SUM (f,FAC,DVFACIN(f,j))) ;
```

```
FORMULA
   ! Initial Share of factor input f  in total costs in industry j !
(all,f,FAC)(all,j,SECT) alphafac(f,j) = DVFACIN(f,j) /
          (SUM(i,SECT,DVCOMIN(i,j)) + SUM (ff,FAC,DVFACIN(ff,j))) ;


FORMULA & EQUATION Comin
    # Intermediate input of commodity i to industry j #
(all,i,SECT)(all,j,SECT) XC(i,j) = DVCOMIN(i,j) / PC(i) ;


FORMULA & EQUATION Facin # Factor input f to industry j #
(all,f,FAC)(all,j,SECT) XF(f,j) = DVFACIN(f,j) / PF(f) ;


FORMULA & EQUATION House # Household demand for commodity i #
(all,i,SECT) XH(i) = DVHOUS(i) / PC(i) ;


FORMULA & EQUATION Com_clear # Commodity market clearing #
(all,i,SECT) XCOM(i) = XH(i) + SUM(j,SECT,XC(i,j)) ;


FORMULA & EQUATION Factor_use # Aggregate primary factor usage #
(all,f,FAC) XFAC(f) = SUM(j,SECT,XF(f,j)) ;



EQUATION(LINEAR)  Consumer_demands # Household expenditure functions #
(all,i,SECT) p_XH(i) = p_Y - p_PC(i) ;


EQUATION(LINEAR)  Intermediate_com
  # Intermediate demands for commodity i by industry j #
(all,i,SECT)(all,j,SECT) p_XC(i,j) = p_XCOM(j) - (p_PC(i) - p_PC(j)) ;


EQUATION(LINEAR)  Factor_inputs    # Factor input demand functions #
(all,f,FAC)(all,j,SECT) p_XF(f,j) = p_XCOM(j) - (p_PF(f) - p_PC(j)) ;


EQUATION(LINEAR) Price_formation # Unit cost index for industry j #
(all,j,SECT) p_PC(j) = SUM(i,SECT,alphacom(i,j)*p_PC(i)) +
                                  SUM(f,FAC,alphafac(f,j)*p_PF(f)) ;

EQUATION(LINEAR) Numeraire
 p_PC("il") = 0 ;


!--------- end of file --------------------!
```

```
!-------------------------------------------------------------!
!              Mixed TABLO Input file for the                 !
!                  Miniature ORANI model                      !
!                                                             !
!          Description of the Miniature ORANI model is in     !
!                sections 4 to 9 of the text                  !
!    "ORANI: A Multisectoral Model of the Australian Economy" !
!      by P.Dixon, B.Parmenter, J.Sutton and D.Vincent        !
!      published by North-Holland 1982 referred to as DPSV.   !
!                                                             !
!    This version is as described in Working Paper No. IP-61  !
!    "Solving Applied General Equilibrium Models represented  !
!      as a Mixture of Linearized and Levels Equations"       !
!      by W.Jill Harrison, K.R.Pearson, Alan A.Powell and     !
!      E.John Small - September 1993 - referred to as HPPS.   !
!-------------------------------------------------------------!

! Text between exclamation marks are comments                 !
! Text between hashes (#) is labelling information            !

!-------------------------------------------------------------!
! DEFAULTS                                                    !
!-------------------------------------------------------------!

EQUATION(DEFAULT=LEVELS) ;
VARIABLE(DEFAULT=LEVELS) ;
VARIABLE(DEFAULT=PERCENT_CHANGE) ;
FORMULA(DEFAULT=INITIAL) ;
COEFFICIENT(DEFAULT=NON_PARAMETER) ;

!-------------------------------------------------------------!
! SETS                                                        !
!-------------------------------------------------------------!

SET COM # commodities # SIZE 2 ;
SET IND # industries  # SIZE 2 ;
SET SOURCE  # source of commodities #  ( domestic, imported ) ;
SET FAC # primary factors #    ( labor, capital ) ;

!-------------------------------------------------------------!
! FILES                                                       !
!-------------------------------------------------------------!

FILE  basedata   # the file containing all base data # ;

!-------------------------------------------------------------!
! VARIABLES     These are VARIABLE(LEVELS)                    !
!-------------------------------------------------------------!

VARIABLE


        (all,i,COM)(all,s,SOURCE)
XHOUS(i,s)    # household consumption #
        ! household consumption of commodity i from source s
           - DPSV X(is)3  ! ;

CHOUS    # nominal total household consumption #
    ! total household consumption of commodities  -  DPSV C ! ;

        (all,i,COM)(all,s,SOURCE)
PCOM(i,s)  # commodity prices #
   ! price (local $) of commodity i from source s - DPSV P(is)! ;

        (all,i,COM)
PDOT(i)  # Unit cost of Armington aggregate i - DPSV P(i.)3 # ;

        (all,i,COM)
XDOT(i) # Armington aggregate over source of commod.i X(i.)3 # ;
```

```
R        # Consumption mix ratio (outer nest - household) # ;

         (all,i,COM)
PEXP(i)  # export prices (foreign dollars) #
         ! price (foreign dollars) of exported commodity i
           - DPSV  p*(i1)  !  ;

         (all,i,COM)
PIMP(i)  # import prices (foreign dollars) #
         ! price (foreign dollars) of imported commodity i
           - DPSV  p*(i2)  !  ;

         (all,i,COM)
XEXP(i)  # export demands #
         ! demand for exported commodity i  - DPSV  x(i1)4  !  ;

         (all,i,COM)
FEXP(i)  # export demand shifters #
  ! shift in demand for exported commodity i - DPSV  f(i1)4! ;

         (all,j,IND)
Z(j)     # industry activity #
         !  activity of industry j    - DPSV Z(j)   !   ;

         (all,i,COM)(all,j,IND)
YCOMIND(i,j)  # industry output #
  !  output of commodity i by industry j   - DPSV y(i1)j  !  ;

         (all,i,COM)(all,s,SOURCE)(all,j,IND)
XINTCOM(i,s,j)  # intermediate commodity inputs #
 !intermediate input of commodity i from source s to industry j
            DPSV x(is)j   for i=1,2  !   ;

         (all,f,FAC)(all,j,IND)
XINTFAC(f,j) # intermediate factor inputs #
          ! intermediate input of factor f to industry j
            - DPSV  x(3f)j ! ;

PLAB   # wage rate #
         !   price of labor (wage rate)  - DPSV p(31)  !  ;

         (all,j,IND)
PCAP(j)   # price of capital #
   !   price of capital in industry j   - DPSV p(32)j   !  ;

         (all,j,IND)(all,f,FAC)
PFAC(f,j) # proce of primary factors #
           ! PLAB and PCAP(j)  ! ;

         (all,i,COM)(all,j,IND)
XINTCOM_CD(i,j)
 #Cobb-Douglas combination of inputs of com i from all sources # ;

         (all,i,COM)(all,j,IND)
PCOM_CD(i,j) # Price index for Cobb-D combination of com i. # ;

         (all,j,IND)
XINTFAC_CD(j)
         #Cobb-Douglas combination of primary factor inputs # ;

         (all,j,IND)
PFAC_CD(j) # Price index for combination of factors #;

         (all,i,COM)
V(i)    # power of export subsidy #
         ! One plus rate of export subsidy for commodity i
           - DPSV  v(i) ! ;

         (all,i,COM)
```

```
T(i)    # power of import duty #
         ! One plus rate of import duty on commodity i
           - DPSV  t(i)  ! ;

XLAB    # total demand for labor #
         !   total demand for labor    - DPSV 1  ("el")    ! ;

         (all,j,IND)
XCAP(j)  # industry demand for capital #
     !   demand for capital in industry j    - DPSV k(j)    ! ;

M        # total imports (foreign or overseas value) #
         ! total imports, overseas (c.i.f. or pre-duty) values
            - DPSV m  ! ;

E        # total exports (foreign or overseas value) #
  !total exports, overseas (f.o.b. or post-subsidy) values
       - DPSV e ! ;

CPI      # consumer price index #    ! DPSV  cpi !  ;

FWAGE    # wage rate shifter #     ! DPSV  f(31) !  ;

CR       # real household consumption #    ! DPSV  cR !  ;

         ! Next variable is added to facilitate the update !
         (all,i,COM)
XIMP(i)  # import quantities # ! DPSV x(i2) ! ;



         (all,i,COM)(all,s,SOURCE)(all,j,IND)
INTCOM(i,s,j) ! intermediate input of commodity i from source s
                 to industry j -  DPSV  P(is).X(is)j     !  ;

         (all,f,FAC)(all,j,IND)
INTFAC(f,j)  ! intermediate input of factor f to industry j  -
              DPSV   P(3f).X(3f)j       !  ;

         (all,i,COM)(all,s,SOURCE)
HOUSE(i,s)  !household consumption of commodity i from source s -
             DPSV P(is).X(is)3 ! ;

         (all,i,COM)
EXPCOM_DOMV(i)  ! Exports of (domestic) commodity i, domestic
          values (that is, pre-subsidy) - DPSV P(i1).X(i1)4  ! ;

         (all,i,COM)
IMPCOM_DOMV(i)  ! Total imports of commodity i (measured in
            Aust dollars) - DPSV P(i2).X(i2) !  ;

         (all,i,COM)(all,j,IND)
COMPROD(i,j)    ! production of commodity i by industry j
             - DPSV  P(i1).Y(i1)j   ! ;

PHI       # exchange rate #
          !   exchange rate ($local/$foreign)
             - DPSV Greek letter phi ! ;


!-------------------------------------------------------------!
! CHANGE VARIABLES                                            !
!-------------------------------------------------------------!

VARIABLE (CHANGE)  B_A   # Aust dollar change in trade balance #
     ! change in balance of trade, overseas values, measured in
          Australian dollar equivalent
           (This is not a percentage change) !
  ! Compare with delB given in DPSV equation (5.57)
  The delB there really has units in foreign dollars.
```

```
 But the implementation of (5.57) has delB measured
 in Australian dollars - see the remarks
 in DPSV on page 32 just after equation (5.57)      !  ;

 ! Next variable is the foreign dollar movement in balance of trade !
VARIABLE (CHANGE)  B_F   # Foreign dollar change in trade balance #
    ! change in balance of trade, overseas values, measured in
       foreign dollars. (This is not a percentage change.) ! ;

VARIABLE(CHANGE) (all,i,COM)   EXPSUB(i)
     ! Export subsidies on exports of commodity i (measured in
       Aust dollars) - DPSV P(i1).X(i1)4 - P*(i1).PHI.X(i1)4 ! ;

VARIABLE(CHANGE) (all,i,COM)   DUTY(i)
      ! Duty levied on imports of commodity i (measured in
       Aust dollars) - DPSV P(i2).X(i2) - P*(i2).PHI.X(i2) !  ;

!-------------------------------------------------------------------!
! PARAMETERS                                                        !
!-------------------------------------------------------------------!

COEFFICIENT(PARAMETER)  (all,i,COM)   GAMMA(i)
       ! the reciprocal of the foreign elasticity of demand for
         exported  commodity i   -   DPSV greek gamma(i)  !
       ! A true parameter - not updated !  ;


COEFFICIENT(PARAMETER)   HWAGE
       ! Wage indexation user-specified parameter - DPSV h ! ;
       ! Usual value is 1.0 - full indexation to cpi.      !


!-------------------------------------------------------------------!
! Levels Variables - Read in for initial solution                   !
!-------------------------------------------------------------------!

READ (all,i,COM)(all,j,IND) INTCOM(i,"domestic",j)
     FROM FILE basedata HEADER "IDOM";

READ (all,i,COM)(all,j,IND) INTCOM(i,"imported",j)
     FROM FILE basedata HEADER "IIMP" ;

READ INTFAC  FROM FILE basedata HEADER "IFAC"    ;

READ (all,i,COM) HOUSE(i,"domestic")
                   FROM FILE basedata HEADER "DOMH" ;
READ (all,i,COM) HOUSE(i,"imported")
                   FROM FILE basedata HEADER "IMPH" ;

READ EXPCOM_DOMV   FROM FILE basedata HEADER "EXP "    ;

READ EXPSUB  FROM FILE basedata HEADER "EXSB" ;

READ DUTY    FROM FILE basedata HEADER "DUTY"    ;

READ COMPROD FROM FILE basedata HEADER "PROD"    ;

READ GAMMA   FROM FILE basedata HEADER "GAMM"    ;

READ PHI   FROM FILE basedata HEADER "PHI " ;

!-------------------------------------------------------------------!
! Calculate Initial Solution - these are FORMULA(INITIAL)           !
!-------------------------------------------------------------------!

FORMULA
       (all,i,COM)(all,s,SOURCE) PCOM(i,s)=1.0 ;

       PLAB = 1.0 ;
```

```
        (all,j,IND) PCAP(j) = 1.0 ;

        CHOUS = SUM(i,COM,SUM(s,SOURCE,HOUSE(i,s))) ;

        (all,j,IND) Z(j)=1.0 ;

        HWAGE = 1.0 ;

        CPI = 1.0 ;


! FORMULA & EQUATIONS   Quantity = Dollar values / Price !
! ------------------------------------------------------ !

FORMULA & EQUATION INTERCOM
        (all,i,COM) (all,s,SOURCE) (all,j,IND)
        XINTCOM(i,s,j) = INTCOM(i,s,j) / PCOM(i,s) ;

FORMULA & EQUATION INTERLAB
        (all,j,IND)
        XINTFAC("labor",j)= INTFAC("labor",j)/PLAB ;

FORMULA & EQUATION INTERCAP
  (all,j,IND)
    XINTFAC("capital",j)=INTFAC("capital",j)/PCAP(j);

FORMULA & EQUATION HOUSEHOLD
        (all,i,COM)(all,s,SOURCE)
        XHOUS(i,s) = HOUSE(i,s) / PCOM(i,s) ;

FORMULA & EQUATION EXPORTCOM
        (all,i,COM) XEXP(i)=EXPCOM_DOMV(i)/PCOM(i,"domestic") ;

FORMULA & EQUATION PRODCOMMOD
        (all,i,COM)(all,j,IND)
        YCOMIND(i,j) = COMPROD(i,j)/PCOM(i,"domestic") ;

FORMULA & EQUATION IMPORTDOMV ! Total Imports !
        (all,i,COM) IMPCOM_DOMV(i)=
       SUM(j,IND,INTCOM(i,"imported",j)) + HOUSE(i,"imported") ;

FORMULA & EQUATION IMPORT_LEVELS
     (all,i,COM)  XIMP(i) = IMPCOM_DOMV(i)/PCOM(i,"imported") ;

FORMULA & EQUATION DUTYUP
        (all,i,COM) PIMP(i) =
                   (IMPCOM_DOMV(i) - DUTY(i))/(PHI * XIMP(i)) ;

!  MARKET CLEARING OF FACTORS
   -------------------------      !

!  This is done in two parts - first labor then capital     !
FORMULA & EQUATION MARKCLLAB
  ! Market clearing of labor - DPSV (5.46) !
    XLAB = SUM(j,IND, XINTFAC("labor",j) ) ;

FORMULA & EQUATION  MARKCLCAP
! Market clearing of capital in industry j - DPSV (5.47) !
  (all,j,IND) XCAP(j) = XINTFAC("capital",j) ;


! ZERO PURE PROFITS IN IMPORTING EQUATION
  ---------------------------------------      !

FORMULA &  EQUATION ZPPROFIMP
 ! Zero pure profits in importing commodity i - DPSV (5.36)  !
        (all,i,COM)
        T(i) = PCOM(i,"imported") / (PHI * PIMP(i)) ;
```

```
FORMULA & EQUATION EXPORTSUBS
        (all,i,COM)
        PEXP(i) = (EXPCOM_DOMV(i) - EXPSUB(i))/(PHI * XEXP(i)) ;

! ZERO PURE PROFITS IN EXPORTING EQUATION
  ---------------------------------------      !

FORMULA & EQUATION ZPPROFEXP
 ! Zero pure profits in exporting commodity i - DPSV (5.35)  !
        (all,i,COM)
        V(i) = PCOM(i,"domestic")/(PEXP(i) * PHI) ;

!  WAGE INDEXATION
   --------------          !

FORMULA & EQUATION WAGEINDEXATION ! DPSV (5.53) !
        FWAGE = PLAB/CPI^HWAGE ;

! AGGREGATE IMPORTS
  -----------------       !

FORMULA & EQUATION IMPORTS
 ! Aggregate imports foreign $ DPSV -(5.49) !
  M = SUM(i,COM,PIMP(i)*XIMP(i)) ;

! AGGREGATE EXPORTS
  -----------------       !

FORMULA & EQUATION EXPORTS
! Aggregrate exports foreign $ - DPSV(5.50) !
  E = SUM(i,COM,PEXP(i)*XEXP(i)) ;

!  CHANGE IN BALANCE OF TRADE
   --------------------------     !

FORMULA & EQUATION TRADEBALANCE_A
   ! Change in balance of trade (overseas
        value)  measured in Aust dollars - cf DPSV (5.57) !
      B_A = (E - M)*PHI ;
FORMULA & EQUATION TRADEBALANCE_F
   ! Change in balance of trade
              measured inforeign dollars - cf DPSV (5.57) !
      B_F= E - M ;

!  REAL HOUSEHOLD CONSUMPTION
   -------------------------       !

FORMULA & EQUATION  REALCONSUMPTION
! Real household consumption -  DPSV (5.54) !
        CR = CHOUS / CPI ;

! EXPORT EQUATION
  --------------  !

FORMULA & EQUATION EXPORTDEMAND
      ! Export demand for commodity i - Levels DPSV (5.18) !
        (all,i,COM) FEXP(i) = PEXP(i) * XEXP(i)^GAMMA(i) ;

!-----------------------------------------------------------------!
!  EQUATIONS -  LINEAR and LEVELS                                 !
!-----------------------------------------------------------------!

! HOUSEHOLD CONSUMPTION
  --------------------- !

  ! See Table 3.1 of HPPS  for more details !
COEFFICIENT(PARAMETER) (all,i,COM)(all,s,SOURCE) ALPHA3(i,s)
   ! share of good (is) in household's total expenditure on commodity i
     -    DPSV Greek alpha(is)3 as defined in DPSV (5.5) ! ;

FORMULA(INITIAL) (all,i,COM)(all,s,SOURCE)
 &     ALPHA3(i,s) = HOUSE(i,s)/SUM(ss,SOURCE,HOUSE(i,ss))  ;
```

```
COEFFICIENT(PARAMETER)
 (all,i,COM) ADOT(i) ! Initial value of Armington aggregate XDOT! ;

FORMULA  ! Equation 3.7 in HPPS !
(all,i,COM)
 PDOT(i) = EXP(SUM(s,SOURCE,ALPHA3(i,s)*LOGE(PCOM(i,s)/ALPHA3(i,s))));

FORMULA  ! Equation 3.8 in HPPS !
(all,i,COM) XDOT(i) = SUM(s,SOURCE,HOUSE(i,s))/PDOT(i) ;

FORMULA (all,i,COM) ADOT(i) = XDOT(i) ;

FORMULA R = 1 ;

EQUATION LEONTIEF  ! Equation 3.4 in HPPS !
(all,i,COM) XDOT(i)/ADOT(i) = R ;

EQUATION BUDGET    ! Equation 3.9 in HPPS !
CHOUS = SUM(i,COM,PDOT(i)*XDOT(i)) ;

EQUATION(LINEAR) E_PDOT ! Equation 3.7** in HPPS !
(all,i,COM)
  p_PDOT(i)= SUM(s,SOURCE,ALPHA3(i,s)*p_PCOM(i,s)) ;

EQUATION(LINEAR) E_XDOT ! Equation 3.5** in HPPS !
(all,i,COM)(all,s,SOURCE)
p_XHOUS(i,s) = p_XDOT(i) - (p_PCOM(i,s) - p_PDOT(i)) ;


! INDUSTRY OUTPUTS EQUATION
  ------------------------  !

COEFFICIENT  (all,q,COM)(all,j,IND)  REVSH(q,j)
  ! Share of commodity q in total production of industry j
        - DPSV R(q1)j ! ;
FORMULA(ALWAYS)    (all,q,COM)(all,j,IND)
   REVSH(q,j) = COMPROD(q,j)/SUM(i,COM,COMPROD(i,j))  ;

EQUATION(LINEAR)  INDOUTPUT
   ! Output of commodity i by industry j - DPSV (5.29)   !
  (all,i,COM)(all,j,IND)
    p_YCOMIND(i,j) = p_Z(j) + p_PCOM(i,"domestic")
          - SUM(q,COM,REVSH(q,j)*p_PCOM(q,"domestic"))  ;

!  INTERMEDIATE USE OF COMMODITIES AND PRIMARY FACTORS
   ------------------------------------------------     !

! Nested Cobb-Douglas forms !
 ! Outer nest is Leontief for effective commodities and
     primary factors !

COEFFICIENT(PARAMETER)  (all,i,COM)(all,j,IND)
    AINTCOM(i,j)  ! A(i.)j - DPSV 5.30 ! ;

COEFFICIENT(PARAMETER)  (all,j,IND)
    AINTFAC(j)    ! A(3.)j - DPSV 5.30 ! ;

EQUATION(LEVELS) Leontief_COM (all,j,IND)(all,i,COM)
   XINTCOM_CD(i,j)/AINTCOM(i,j) = Z(j) ;

EQUATION(LEVELS) Leontief_FAC (all,j,IND)
   XINTFAC_CD(j)/AINTFAC(j) = Z(j) ;

! COMMODITIES
  -----------      !

COEFFICIENT(PARAMETER)   (all,i,COM)(all,s,SOURCE)(all,j,IND)
                              ALPHACOM(i,s,j)
  !  Share of good (is) in total use of commodity i by
  industry j -  DPSV  Greek alpha(is)j as defined in DPSV (5.31)!;
```

```
FORMULA(INITIAL)    (all,i,COM)(all,s,SOURCE)(all,j,IND)
  ALPHACOM(i,s,j) = INTCOM(i,s,j)/SUM(ss,SOURCE,INTCOM(i,ss,j)) ;

  ! Cobb-Douglas combination over domestic and imported sources !
EQUATION(LINEAR) INTUSECOM_P (all,j,IND)(all,i,COM)
p_PCOM_CD(i,j)= SUM(s,SOURCE,ALPHACOM(i,s,j)*p_PCOM(i,s)) ;

EQUATION(LINEAR) INTUSECOM_X (all,j,IND)(all,i,COM)(all,s,SOURCE)
p_XINTCOM(i,s,j)= p_XINTCOM_CD(i,j) - (p_PCOM(i,s) - p_PCOM_CD(i,j)) ;

! LABOR AND CAPITAL
  ----------------              !

COEFFICIENT(PARAMETER)    (all,f,FAC)(all,j,IND) ALPHAFAC(f,j)
  !  Share of factor f in total use of primary factors by industry j
      -  DPSV  Greek alpha(3f)j as defined in DPSV (5.31)    !  ;
FORMULA(INITIAL)    (all,f,FAC)(all,j,IND)
      ALPHAFAC(f,j) = INTFAC(f,j)/SUM(g,FAC,INTFAC(g,j))    ;

! Cobb-Douglas combination over labor and capital !
EQUATION E_PFAC_LAB (all,j,IND)
      PFAC("labor",j) = PLAB ;
EQUATION E_PFAC_CAP (all,j,IND)
      PFAC("capital",j) = PCAP(j) ;

EQUATION(LINEAR) INTUSEFAC_P (all,j,IND)
p_PFAC_CD(j)= SUM(f,FAC,ALPHAFAC(f,j)*p_PFAC(f,j)) ;

EQUATION(LINEAR) INTUSEFAC_X (all,j,IND)(all,f,FAC)
p_XINTFAC(f,j)= p_XINTFAC_CD(j) - (p_PFAC(f,j) - p_PFAC_CD(j)) ;

!  CONSUMER PRICE INDEX
   --------------------          !

COEFFICIENT (all,i,COM)(all,s,SOURCE) SHOUS(i,s)
  ! share of total household budget devoted to commodity i
     from source s    -     DPSV S(is)3  !    ;

FORMULA(ALWAYS) (all,i,COM)(all,s,SOURCE)
    SHOUS(i,s) = HOUSE(i,s)/SUM(q,COM,SUM(ss,SOURCE,HOUSE(q,ss))) ;

EQUATION(LINEAR)    CONSPRICEINDEX
     ! Consumer price index - DPSV (5.58) !
   p_CPI = SUM(i,COM, SUM(s, SOURCE, SHOUS(i,s)*p_PCOM(i,s)) )   ;


!-------------------------------------------------------------------!
!     EQUATION(LEVELS)  - must hold with initial database           !
!        since not used as initial formula                          !
!-------------------------------------------------------------------!

!   ZERO PURE PROFITS IN PRODUCTION EQUATION
    ---------------------------------------        !
EQUATION ZPPROFPROD
 ! Zero pure profits in production of each industry - DPSV (5.34) !
   (all,j,IND) SUM(i,COM,COMPROD(i,j))=
    SUM(i,COM, SUM(s,SOURCE, INTCOM(i,s,j) ) )
                  + SUM(f,FAC, INTFAC(f,j) )   ;

!  MARKET CLEARING OF COMMODITIES
   ------------------------------       !

EQUATION MARKCLCOM
  ! Market clearing of commodity i - DPSV (5.44) !
   (all,i,COM) SUM(j,IND, YCOMIND(i,j) ) =
         SUM(j,IND,XINTCOM(i,"domestic",j)) +
                     XHOUS(i,"domestic") + XEXP(i) ;
!-------------------------------------------------------------------!
```

```
!-------------------------------------------------------------!
!           Mixed TABLO Input file for the ORANI-F model      !
!                  as described in the article                !
! "ORANI-F:A General Equilibrium Model of the Australian Economy" !
!       by J.M.Horridge, B.R.Parmenter, and K.R.Pearson [HPP] !
!     published the journal "Economic and Financial Computing", !
!                 volume 3, 1993, pages 71-140.               !
!                                                             !
!       This version is as described in Working Paper No. IP-61 !
!       "Solving Applied General Equilibrium Models represented !
!         as a Mixture of Linearized and Levels Equations"    !
!         by W.Jill Harrison, K.R.Pearson, Alan A.Powell and  !
!         E.John Small -[HPPS] September 1993.                !
!-------------------------------------------------------------!

! Excerpts refer to Excerpts as described in the above paper HPP !

! DEFAULTS !
!.........!

EQUATION (DEFAULT=LEVELS);
VARIABLE (DEFAULT=LEVELS);
VARIABLE (DEFAULT=PERCENT_CHANGE);
FORMULA (DEFAULT=INITIAL);
COEFFICIENT (DEFAULT=NON_PARAMETER);

! Excerpt 1 of TABLO input file: !
!...............................!

! Definitions of sets !

Set
! Subscript !
COM         # Commodities #              (C1 - C23);                ! c !
SRC         # Source of Commodities #    (dom,imp);                 ! s !
IND         # Industries #                (I1 - I22);                ! i !
OCC         # Occupation Types #          (skilled,unskilled);       ! o !
MAR         # Margin Commodities #       (C18,C19);                 ! m !
NONMAR      # Non-Margin Commodities #   (C1 - C17, C20 - C23);     ! n !
Subset MAR        is subset of COM;
NONMAR      is subset of COM;

! Excerpt 2 of TABLO input file: !
!...............................!

! Variables relating to commodity flows !

Variable
! Basic Demands for commodities (excluding margin demands) !
(All,c,COM)(All,s,SRC)(All,i,IND)  x1(c,s,i)  # Intermediate #;
(All,c,COM)(All,s,SRC)(All,i,IND)  x2(c,s,i)  # Investment #;
(All,c,COM)(All,s,SRC)             x3(c,s)    # Household #;
(All,c,COM)                        x4(c)      # Export #;
(All,c,COM)(All,s,SRC)             x5(c,s)    # Other #;
(change)
(All,c,COM)                        x6(c)      # Inventories #;

(All,c,COM)(All,s,SRC) p0(c,s)
                    # basic price of commodity c, source s #;

! Technical or Taste Change Variables affecting Basic Demands !
(All,c,COM)(All,s,SRC)(All,i,IND)  a1(c,s,i)  # Intermediate #;
(All,c,COM)(All,s,SRC)(All,i,IND)  a2(c,s,i)  # Investment #;
(All,c,COM)(All,s,SRC)             a3(c,s)    # Household #;
(All,c,COM)(All,s,SRC)             f5(c,s)    # Other Demand Shift #;

! Margin Usage on Basic Flows !
(All,c,COM)(All,s,SRC)(All,i,IND)(All,m,MAR)  x1mar(c,s,i,m)
```

```
# Intermediate #;
(All,c,COM)(All,s,SRC)(All,i,IND)(All,m,MAR)   x2mar(c,s,i,m)
# Investment #;
(All,c,COM)(All,s,SRC)(All,m,MAR)              x3mar(c,s,m)
# Household #;
(All,c,COM)(All,m,MAR)                         x4mar(c,m)
# Export #;
(All,c,COM)(All,s,SRC)(All,m,MAR)              x5mar(c,s,m)
# Other #;

! Technical Change in Margins Usage !
(All,c,COM)(All,s,SRC)(All,i,IND)(All,m,MAR)   a1mar(c,s,i,m)
# Intermediate #;
(All,c,COM)(All,s,SRC)(All,i,IND)(All,m,MAR)   a2mar(c,s,i,m)
# Investment #;
(All,c,COM)(All,s,SRC)(All,m,MAR)              a3mar(c,s,m)
# Household #;
(All,c,COM)(All,m,MAR)                         a4mar(c,m)
# Export #;
(All,c,COM)(All,s,SRC)(All,m,MAR)              a5mar(c,s,m)
# Other #;

! Powers of Commodity Taxes on Basic Flows !
(All,c,COM)(All,s,SRC)(All,i,IND)   t1(c,s,i)  # Intermediate #;
(All,c,COM)(All,s,SRC)(All,i,IND)   t2(c,s,i)  # Investment #;
(All,c,COM)(All,s,SRC)              t3(c,s)    # Household #;
(All,c,COM)                         t4(c)      # Export #;
(All,c,COM)(All,s,SRC)              t5(c,s)    # Other #;

! Purchaser's Prices (including margins and taxes) !
(All,c,COM)(All,s,SRC)(All,i,IND)   p1(c,s,i)  # Intermediate #;
(All,c,COM)(All,s,SRC)(All,i,IND)   p2(c,s,i)  # Investment #;
(All,c,COM)(All,s,SRC)              p3(c,s)    # Household #;
(All,c,COM)                         p4(c)      # Exports $A #;
(All,c,COM)(All,s,SRC)              p5(c,s)    # Other #;

! Excerpt 3 of TABLO input file: !
!.............................!

!  Variables for primary-factor flows, commodity supplies
        and import duties !

! Variables relating to usage of labour, occupation o, in industry i !
(All,i,IND)(All,o,OCC) x1lab(i,o) # Employment #;
(All,i,IND)(All,o,OCC) p1lab(i,o) # Wage #;
(All,i,IND)            a1lab_o(i) # Labor Augmenting Technical Change #;
(All,i,IND)(All,o,OCC) f1lab(i,o) # Wage Shift Variable #;

! Variables relating to usage of fixed capital in industry i !
(All,i,IND)  x1cap(i)     # Current Capital Stock #;
(All,i,IND)  p1cap(i)     # Rental Price of Capital #;
(All,i,IND)  a1cap(i)     # Capital Augmenting Technical Change #;
(All,i,IND)  r1cap(i)       # Current Rates of Return on Fixed Capital #;

! Variables relating to usage of land !
(All,i,IND)  x1lnd(i)        # Use of Land #;
(All,i,IND)  p1lnd(i)        # Rental Price of Land #;
(All,i,IND)  a1lnd(i)        # Land Augmenting Technical Change #;

! Variables relating to "Other Costs" !
(All,i,IND)  x1oct(i) # Demand for "Other Cost" Tickets #;
(All,i,IND)  p1oct(i) # Price of "Other Cost" Tickets #;
(All,i,IND)  a1oct(i) # "Other Cost" Ticket Augmenting Techncal Change#;
(All,i,IND)  f1oct(i) # Shifts in Price of "Other Cost" Tickets #;

! Variables relating to commodity supplies and import duties !
(All,c,COM)(All,i,IND) q1(c,i)  # Output of commodity c by industry i #;
(All,c,COM)            t0imp(c) # Power of Tariffs #;
```

```
! Excerpt 4 of TABLO input file: !
!.............................!

! Variables describing composite commodities !

! Demands for import/domestic commodity composites !
(All,c,COM)(All,i,IND) x1_s(c,i) # Intermediate #;
(All,c,COM)(All,i,IND) x2_s(c,i) # Investment #;
(All,c,COM)            x3_s(c)   # Household #;
(All,c,COM)            x3lux(c)  # Household - Supernumerary Demands #;
(All,c,COM)            x3sub(c)  # Household - Subsistence Demands #;

! Effective Prices of import/domestic commodity composites !
(All,c,COM)(All,i,IND)  p1_s(c,i)  # Intermediate #;
(All,c,COM)(All,i,IND)  p2_s(c,i)  # Investment #;
(All,c,COM)             p3_s(c)    # Household #;
! Technical or Taste Change Variables for import/domestic composites !
(All,c,COM)(All,i,IND)  a1_s(c,i) # Intermediate #;
(All,c,COM)(All,i,IND)  a2_s(c,i) # Investment #;
(All,c,COM)             a3_s(c)   # All Household Usage of Good c #;
(All,c,COM)             a3lux(c)  # Household - Supernumerary Demands #;
(All,c,COM)             a3sub(c)  # Household - Subsistence Demands #;

! Excerpt 5 of TABLO input file: !
!.............................!
! Miscellaneous vector variables !
Variable
(All,c,COM)  f0tax_s(c)      # General Sales Tax Shifter #;
(All,c,COM)  f4p(c)   # Price (upward) Shift in Export Demand Schedule#;
(All,c,COM)  f4q(c)        # Quantity (right) Shift in Export Demands #;
(All,c,COM)  pf0cif(c)       # C.I.F. Foreign Currency Import Prices #;
(All,c,COM)  x0dom(c)        # Total Supplies of Domestic Goods #;
(All,c,COM)  x0imp(c)        # Total Supplies of Imported Goods #;
(All,i,IND)  alprim(i)       # All Factor Augmenting Technical Change #;
(All,i,IND)  altot(i)        # All Input Augmenting Technical Change #;
(All,i,IND)  a2tot(i)        # Neutral Technical Change - Investment #;
(All,i,IND)  employ(i)       # Employment by Industry #;
(All,i,IND)  f1lab_o(i)      # Industry-Specific Wage Shifter #;
(All,i,IND)  f_accum(i)      # Capital Accumulation Shifter #;
(All,i,IND)  f1ret(i)        # Rate of Return Shifter #;
(All,i,IND)  p1lab_o(i)      # Price of Labour Composite #;
(All,i,IND)  p1prim(i)  # Effective Price of Primary Factor Composite #;
(All,i,IND)  p1tot(i)        # Average Input/Output Price #;
(All,i,IND)  p2tot(i)        # Costs of Units of Capital #;
(All,i,IND)  x1lab_o(i)      # Effective Labour Input #;
(All,i,IND)  x1prim(i)       # Primary Factor Composite #;
(All,i,IND)  x1tot(i)        # Activity Level or Value-Added #;
(All,i,IND)  x2tot(i)        # Investment by Using Industry #;
(All,o,OCC)  f1lab_i(o)      # Occupation-Specific Wage Shifter #;
(All,o,OCC)  x1lab_i(o)      # Employment by Occupation #;

! Excerpt 6 of TABLO input file: !
!.............................!
! Scalar or macro variables !
Variable
(change)
B              # Balance of Trade #;
(change)
BT             # Real Trade Deficit #;
(change)
Fudge          # "Fudge Factor": set to Unity for dynamic simulation #;
Variable
employ_i  # Aggregate Employment- Wage Bill Weights #;
f1lab_io  # Overall Wage Shifter #;
f1tax_csi # Uniform % Change in Powers of Taxes on Intermediate Usage #;
f2tax_csi # Uniform % Change in Powers of Taxes on Investment #;
f3tax_cs  # Uniform % Change in Powers of Taxes on Household Usage #;
```

```
f4_ntrad  # Demand Shift, Non-Traditional Export Aggregate #;
f4tax_ntrad # Uniform % Change in Powers of Taxes on NonTradl Exports #;
f4tax_trad  # Uniform % Change in Powers of Taxes on Tradtnl Exports #;
f5tax_cs     # Uniform % Change in Powers of Taxes on "Other" Usage #;
f5tot        # Overall Shift Term For "Other" Demands #;
f5tot2       # Ratio between f5tot and p_x3tot #;
p0cif_c      # Imports Price Index, CIF, A$ #;
p0gdpexp     # GDP Price Index, Expenditure Side #;
p0imp_c      # Duty-paid Imports Price Index,A$ #;
p0realdev    # Real Devaluation #;
p0toft       # Terms of Trade #;
p1cap_i      # Average Capital Rental #;
p2tot_i      # Aggregate Investment Price Index #;
p3tot        # Consumer Price Index #;
p4_ntrad     # Price, Non-Traditional Export Aggregate #;
p4tot        # Exports Price Index #;
p5tot        # "Other" Demands Price Index #;
p6tot        # Inventories Price Index #;
phi          # Exchange Rate, $A/$world #;
q            # Number of Households #;
r1cap_i      # Average Rate of Return #;
utility      # Utility per Household #;
w0cif_c      # CIF A$ Value of Imports #;
w0gdpexp     # Nominal GDP from Expenditure Side #;
w0imp_c      # Value of Imports plus Duty #;
w0tar_c      # Aggregate Tariff Revenue #;
w1tax_csi    # Aggregate Revenue from Indirect Taxes on Intermediate #;
w2tax_csi    # Aggregate Revenue from Indirect Taxes on Investment #;
w2tot_i      # Aggregate Nominal Investment #;
w3lux        # Total Nominal Supernumerary Household Expenditure #;
w3tax_cs     # Aggregate Revenue from Indirect Taxes on Households #;
w3tot        # Nominal Total Household Consumption #;
w4tax_c      # Aggregate Revenue from Indirect Taxes on Export #;
w4tot        # A$ Border Value of exports #;
w5tax_cs     # Aggregate Revenue from Indirect Taxes on "Other" #;
w5tot        # Aggregate Nominal Value of "Other" Demands #;
w6tot        # Aggregate Nominal Value of Inventories #;
x0cif_c      # Import Volume Index, CIF Weights #;
x0gdpexp     # Real GDP from Expenditure Side #;
x0imp_c      # Import Volume Index, Duty-Paid Weights #;
x1cap_i      # Aggregate Capital Stock, Rental Weights #;
x1prim_i     # Aggregate Output: Value-Added Weights #;
x2tot_i      # Aggregate Real Investment Expenditure #;
x3tot        # Real Household Consumption #;
x4_ntrad     # Quantity, Non-Traditional Export Aggregate #;
x4tot        # Export Volume Index #;
x5tot        # Aggregate Real "Other" Demands #;
x6tot        # Aggregate Real Inventories #;

! Extra levels variables previously coefficients !

! The following variables are made CHANGE variables to avoid
 difficulties with zero values in the database !

VARIABLE(default = change) ;


File  MDATA  # Data File #;


! Excerpt 7 of TABLO input file: !
!..............................!

! Data relating to basic commodity flows !

Variable  ! Basic Flows of Commodities!
 (All,c,COM)(All,s,SRC)(All,i,IND)  V1BAS(c,s,i)  # Intermediate #;
 (All,c,COM)(All,s,SRC)(All,i,IND)  V2BAS(c,s,i)  # Investment   #;
 (All,c,COM)(All,s,SRC)             V3BAS(c,s)    # Households   #;
 (All,c,COM)                        V4BAS(c)      # Export       #;
```

```
 (All,c,COM) (All,s,SRC)                V5BAS(c,s)     # Other Demand #;
 (All,c,COM)                            V6BAS(c)       # Inventories  #;
Read
 V1BAS From File MDATA Header "1BAS";
 V2BAS From File MDATA Header "2BAS";
 V3BAS From File MDATA Header "3BAS";
 V4BAS From File MDATA Header "4BAS";
 V5BAS From File MDATA Header "5BAS";
 V6BAS From File MDATA Header "6BAS";

Zerodivide Default 0.0;
Zerodivide (Nonzero_by_Zero) Default 0.0;
Coefficient (parameter) tiny ;
Formula tiny = 0.0000000001;

Formula ! Initial settings for basic prices !
(All,c,COM) (All,s,SRC) p0(c,s) = 1;

Formula & Equation S_x1 (All,c,COM) (All,s,SRC) (All,i,IND)
  x1(c,s,i) = V1BAS(c,s,i) / p0(c,s);
Formula & Equation S_x2 (All,c,COM) (All,s,SRC) (All,i,IND)
  x2(c,s,i) = V2BAS(c,s,i) / p0(c,s);
Formula & Equation S_x3 (All,c,COM) (All,s,SRC)
  x3(c,s)   = V3BAS(c,s) / p0(c,s) ;
Formula & Equation S_x4 (All,c,COM)
  x4(c)     = V4BAS(c) / p0(c,"dom") ;
Formula & Equation S_x5 (All,c,COM) (All,s,SRC)
  x5(c,s)   = V5BAS(c,s) / p0(c,s) ;
Formula & Equation S_x6 (All,c,COM)
  x6(c)     = V6BAS(c) / p0(c,"dom") ;

Variable  ! Margin Flows!
 (All,c,COM) (All,s,SRC) (All,i,IND) (All,m,MAR)
                                       V1MAR(c,s,i,m) # Intermediate #;
  (All,c,COM) (All,s,SRC) (All,i,IND) (All,m,MAR)
                                       V2MAR(c,s,i,m) # Investment #;
 (All,c,COM) (All,s,SRC) (All,m,MAR)  V3MAR(c,s,m)   # Households #;
  (All,c,COM) (All,m,MAR)             V4MAR(c,m)     # Export #;
  (All,c,COM) (All,s,SRC) (All,m,MAR) V5MAR(c,s,m)   # Other #;
Read
 V1MAR From File MDATA Header "1MAR";
 V2MAR From File MDATA Header "2MAR";
 V3MAR From File MDATA Header "3MAR";
 V4MAR From File MDATA Header "4MAR";
 V5MAR From File MDATA Header "5MAR";

Formula & Equation S_x1mar (All,c,COM) (All,s,SRC) (All,i,IND) (All,m,MAR)
  x1mar(c,s,i,m) = V1MAR(c,s,i,m) / p0(m,"dom");
Formula & Equation S_x2mar (All,c,COM) (All,s,SRC) (All,i,IND) (All,m,MAR)
  x2mar(c,s,i,m) = V2MAR(c,s,i,m) / p0(m,"dom");
Formula & Equation S_x3mar (All,c,COM) (All,s,SRC) (All,m,MAR)
  x3mar(c,s,m)   = V3MAR(c,s,m) / p0(m,"dom");
Formula & Equation S_x4mar (All,c,COM) (All,m,MAR)
  x4mar(c,m)     = V4MAR(c,m) / p0(m,"dom");
Formula & Equation S_x5mar (All,c,COM) (All,s,SRC) (All,m,MAR)
  x5mar(c,s,m)   = V5MAR(c,s,m) / p0(m,"dom");

! Excerpt 8 of TABLO input file: !
!...............................!

! Data relating to commodity taxes !

Variable ! Taxes on Basic Flows!
 (All,c,COM) (All,s,SRC) (All,i,IND)  V1TAX(c,s,i);
 (All,c,COM) (All,s,SRC) (All,i,IND)  V2TAX(c,s,i);
 (All,c,COM) (All,s,SRC)              V3TAX(c,s);
 (All,c,COM)                          V4TAX(c);
 (All,c,COM) (All,s,SRC)              V5TAX(c,s);
```

```
Read
 V1TAX From File MDATA Header "1TAX";
 V2TAX From File MDATA Header "2TAX";
 V3TAX From File MDATA Header "3TAX";
 V4TAX From File MDATA Header "4TAX";
 V5TAX From File MDATA Header "5TAX";

Formula  (All,c,COM)(All,s,SRC)(All,i,IND)
 t1(c,s,i) = V1TAX(c,s,i) / V1BAS(c,s,i) + 1;
Formula (All,c,COM)(All,s,SRC)(All,i,IND)
 t2(c,s,i) = V2TAX(c,s,i) / V2BAS(c,s,i) + 1;
Formula (All,c,COM)(All,s,SRC)
 t3(c,s) = V3TAX(c,s) / V3BAS(c,s) + 1;
Formula (All,c,COM)
 t4(c) = V4TAX(c) / V4BAS(c) + 1;
Formula (All,c,COM)(All,s,SRC)
 t5(c,s) = V5TAX(c,s) / V5BAS(c,s) + 1;

Equation S_t1 (All,c,COM)(All,s,SRC)(All,i,IND)
V1BAS(c,s,i) * t1(c,s,i) = V1TAX(c,s,i) + V1BAS(c,s,i);
Equation S_t2 (All,c,COM)(All,s,SRC)(All,i,IND)
V2BAS(c,s,i) * t2(c,s,i) = V2TAX(c,s,i) + V2BAS(c,s,i);
Equation S_t3 (All,c,COM)(All,s,SRC)
V3BAS(c,s) * t3(c,s) = V3TAX(c,s) + V3BAS(c,s);
Equation S_t4 (All,c,COM)
V4BAS(c) * t4(c) = V4TAX(c) + V4BAS(c);
Equation S_t5 (All,c,COM)(All,s,SRC)
V5BAS(c,s) * t5(c,s) = V5TAX(c,s) + V5BAS(c,s);

! Excerpt 11 of TABLO input file: !
!.............................!

! Aggregates of flows at purchasers' prices !

Variable ! Flows at Purchasers prices !
 (All,c,COM)(All,s,SRC)(All,i,IND)  V1PUR(c,s,i)  # Intermediate #;
 (All,c,COM)(All,s,SRC)(All,i,IND)  V2PUR(c,s,i)  # Investment   #;
 (All,c,COM)(All,s,SRC)             V3PUR(c,s)    # Households   #;
 (All,c,COM)                        V4PUR(c)      # Export       #;
 (All,c,COM)(All,s,SRC)             V5PUR(c,s)    # Other Demand #;

Formula & Equation S_V1PUR (All,c,COM)(All,s,SRC)(All,i,IND)
 V1PUR(c,s,i) = V1BAS(c,s,i) + V1TAX(c,s,i) + Sum(m,MAR,V1MAR(c,s,i,m));
Formula & Equation S_V2PUR (All,c,COM)(All,s,SRC)(All,i,IND)
 V2PUR(c,s,i) = V2BAS(c,s,i) + V2TAX(c,s,i) + Sum(m,MAR,V2MAR(c,s,i,m));
Formula & Equation S_V3PUR (All,c,COM)(All,s,SRC)
  V3PUR(c,s)  = V3BAS(c,s)   + V3TAX(c,s)   + Sum(m,MAR,V3MAR(c,s,m));
Formula & Equation S_V4PUR (All,c,COM)
  V4PUR(c)    = V4BAS(c)     + V4TAX(c)     + Sum(m,MAR,V4MAR(c,m));
Formula & Equation S_V5PUR (All,c,COM)(All,s,SRC)
  V5PUR(c,s)  = V5BAS(c,s)   + V5TAX(c,s)   + Sum(m,MAR,V5MAR(c,s,m));

Variable ! Flows at Purchaser's prices: Domestic + Imported Totals !
 (All,c,COM)(All,i,IND)     V1PUR_S(c,i);
 (All,c,COM)(All,i,IND)     V2PUR_S(c,i);
 (All,c,COM)                V3PUR_S(c);

Formula & Equation S_V1PUR_S # Flows at Purchasers prices #
 (All,c,COM)(All,i,IND)     V1PUR_S(c,i)  = Sum(s,SRC,V1PUR(c,s,i));
Formula & Equation S_V2PUR_S # Flows at Purchasers prices #
 (All,c,COM)(All,i,IND)     V2PUR_S(c,i)  = Sum(s,SRC,V2PUR(c,s,i));
Formula & Equation S_V3PUR_S # Flows at Purchasers prices #
 (All,c,COM)                V3PUR_S(c)    = Sum(s,SRC,V3PUR(c,s));


! Imports and tariffs !

Variable
 (All,c,COM) V0IMP(c)     # Total basic-value imports of good c #;
 (All,c,COM) V0TAR(c)     # tariff revenue #;
```

```
(All,c,COM) V0CIF(c)      # Total ex-duty imports of good c# ;

Read V0TAR From File MDATA Header "0TAR";

Formula & Equation S_V0IMP (All,c,COM) V0IMP(c) =
  Sum(i,IND,V1BAS(c,"imp",i) + V2BAS(c,"imp",i))
 + V3BAS(c,"imp") + V5BAS(c,"imp") ;

Formula & Equation S_V0CIF (All,c,COM)
V0CIF(c) = V0IMP(c) - V0TAR(c);

Formula & Equation S_t0imp (All,c,COM)
 t0imp(c) =  1.0 + V0TAR(c) / V0CIF(c);

! Excerpt 9 of TABLO input file: !
!..............................!

! Data relating to primary-factor flows !

Variable  ! Primary Factor and Other Industry costs!
 (All,i,IND)                V1CAP(i)     # capital rentals #;
 (All,i,IND) (All,o,OCC)    V1LAB(i,o)   # wage bill matrix #;
 (All,i,IND)                V1LND(i)     # land rentals #;
 (All,i,IND)                V1OCT(i)     # other cost tickets #;


Read
 V1CAP From File MDATA Header "1CAP";
 V1LAB From File MDATA Header "1LAB";
 V1LND From File MDATA Header "1LND";
 V1OCT From File MDATA Header "1OCT";

Formula
 (All,i,IND)                p1cap(i)    = 1;
 (All,i,IND) (All,o,OCC)    p1lab(i,o)  = 1;
 (All,i,IND)                p1lnd(i)    = 1;
 (All,i,IND)                p1oct(i)    = 1;

Formula & Equation S_x1cap (All,i,IND)
  x1cap(i) = V1CAP(i) / p1cap(i);
Formula & Equation S_x1lab (All,i,IND) (All,o,OCC)
  x1lab(i,o) = V1LAB(i,o) / p1lab(i,o);
Formula & Equation S_x1lnd (All,i,IND)
  x1lnd(i) = V1LND(i) / p1lnd(i);
Formula & Equation S_x1oct (All,i,IND)
  x1oct(i) = V1OCT(i) / p1oct(i);

! Excerpt 12 of TABLO input file: !
!..............................!

! Cost and usage aggregates !

Variable
(All,i,IND)  V1LAB_0(i)  # total labour bill in industry i #;
(All,i,IND)  V1PRIM(i)   # total factor input to industry i#;
(All,i,IND)  V1TOT(i)    # total cost in each industry #;
(All,i,IND)  V2TOT(i)    # total capital created for each industry #;
(All,o,OCC)  V1LAB_I(o)  # total wages, occupation o #;

Formula & Equation S_V1LAB_0 # Total labour bill in industry i #
 (All,i,IND)  V1LAB_0(i)  = Sum(o,OCC,  V1LAB(i,o));
Formula & Equation S_V1PRIM # Total factor input in industry i #
 (All,i,IND)  V1PRIM(i)   = V1LAB_0(i) + V1CAP(i) + V1LND(i);
Formula & Equation S_V1TOT # Total cost in industry i #
 (All,i,IND)  V1TOT(i)    = Sum(c,COM,  V1PUR_S(c,i)) + V1PRIM(i) + V1OCT(i);
Formula & Equation S_V2TOT # Total capital in industry i #
 (All,i,IND)  V2TOT(i)    = Sum(c,COM,  V2PUR_S(c,i));
Formula & Equation S_V1LAB_I # Total wages occupation o #
 (All,o,OCC)  V1LAB_I(o)  = Sum(i,IND,  V1LAB(i,o));
```

```
! Excerpt 14 of TABLO input file: !
!..............................!

! Expenditure-side components of GDP !

Variable ! Expenditure Aggregates at Purchaser's Prices !
 V0CIF_C  # Total A$ import costs, excluding tariffs #;
 V0IMP_C  # Total basic-value imports (includes tariffs) #;
 V2TOT_I  # Total investment usage #;
 V3TOT    # Total purchases by households #;
 V4TOT    # Total export earnings #;
 V5TOT    # total value of other demands #;
 V6TOT    # total value of inventories #;
 V0GDPEXP # Nominal GDP from expenditure side #;

Formula & Equation S_V0CIF_C
 V0CIF_C   = Sum(c,COM,V0CIF(c));
Formula & Equation S_V0IMP_C
 V0IMP_C   = Sum(c,COM,V0IMP(c));
Formula & Equation S_V2TOT_I
 V2TOT_I   = Sum(i,IND,  V2TOT(i));
Formula & Equation S_V3TOT
 V3TOT     = Sum(c,COM,  V3PUR_S(c));
Formula & Equation S_V4TOT
 V4TOT     = Sum(c,COM,  V4PUR(c));
Formula & Equation S_V5TOT
 V5TOT     = Sum(c,COM,Sum(s,SRC,  V5PUR(c,s)));
Formula & Equation S_V6TOT
 V6TOT     = Sum(c,COM,  V6BAS(c));
Formula & Equation S_V0GDPEXP
 V0GDPEXP  = V3TOT + V2TOT_I + V5TOT + V6TOT + V4TOT - V0CIF_C;

! Excerpt 10 of TABLO input file: !
!..............................!

! Data relating to commodity outputs !

Variable (All,c,COM)(All,i,IND) MAKE(c,i)
# production of commodity c by industry i #;
Read MAKE From File MDATA Header "MAKE";

Formula & Equation S_q1 # Volumes of commodities produced by industry #
  (All,c,COM)(All,i,IND) q1(c,i) = MAKE(c,i) / p0(c,"dom");

! Excerpt 27 of TABLO input file: !
!..............................!

! Market clearing equations !

Formula & Equation E_x0dom # Total output of domestic commodities #
(all,c,COM) x0dom(c) = Sum(i,IND, q1(c,i));

Equation E_p0_B # Aggregate demand for non margin commodities #
(All,n,NONMAR) x0dom(n) =
  Sum(i,IND,x1(n,"dom",i) + x2(n,"dom",i))
  + x3(n,"dom") + x4(n)
  + x5(n,"dom") + x6(n);

Equation E_p0_C # Aggregate demand for margin commodities #
(All,m,MAR)
  x0dom(m) =                ! basic part first !
     Sum(i,IND, x1(m,"dom",i) + x2(m,"dom",i))
  + x3(m,"dom") + x4(m)
  + x5(m,"dom") + x6(m)     ! now margin part !
  + Sum(c,COM, x4mar(c,m)    ! note nesting of Sum parentheses !
  +  Sum(s,SRC, x3mar(c,s,m) + x5mar(c,s,m)
  +   Sum(i,IND, x1mar(c,s,i,m) + x2mar(c,s,i,m))));
```

```
Formula & Equation E_x0imp # Import volumes #
(All,c,COM) x0imp(c) =
    Sum(i,IND, x1(c,"imp",i) + x2(c,"imp",i))
        + x3(c,"imp") + x5(c,"imp");

Formula & Equation E_x1lab_i
# Aggregate Demand for labour of each skill #
(All,o,OCC) x1lab_i(o) = Sum(i,IND, x1lab(i,o));

! Initial values for levels variables !

Formula
phi = 1;

Formula (All,c,COM)
 pf0cif(c) = V0CIF(c) / (phi*x0imp(c));

! Excerpt 11 of TABLO input file: !
!..............................!

! Aggregates and shares of flows at purchasers' prices !

Coefficient ! Source Shares in Flows at Purchaser's prices !
 (All,c,COM)(All,s,SRC)(All,i,IND)  S1(c,s,i);
 (All,c,COM)(All,s,SRC)(All,i,IND)  S2(c,s,i);
 (All,c,COM)(All,s,SRC)             S3(c,s);

Zerodivide off;
Zerodivide Default 0.5;

Formula (Always)
 (All,c,COM)(All,s,SRC)(All,i,IND) S1(c,s,i)=V1PUR(c,s,i)/V1PUR_S(c,i);
 (All,c,COM)(All,s,SRC)(All,i,IND) S2(c,s,i)=V2PUR(c,s,i)/V2PUR_S(c,i);
 (All,c,COM)(All,s,SRC)            S3(c,s)  =V3PUR(c,s)  /V3PUR_S(c);
Zerodivide Off;
Zerodivide default 0.0;

Coefficient (All,c,COM) V0MAR_CSI(c)
         # Total usage for margins purposes #;
Formula (Always) (All,m,MAR) V0MAR_CSI(m) =
      Sum(c,COM, V4MAR(c,m) +
                 Sum(s,SRC, V3MAR(c,s,m) + V5MAR(c,s,m) +
                      Sum(i,IND, V1MAR(c,s,i,m) + V2MAR(c,s,i,m) ) ));
Formula (Always) (All,n,NONMAR) V0MAR_CSI(n) = 0.0;

Coefficient (All,c,COM) SALES(c)
                # Total sales of domestic commodity c #;
Formula (Always) (All,c,COM)
SALES(c) = Sum(i,IND,V1BAS(c,"dom",i) + V2BAS(c,"dom",i))
+ V3BAS(c,"dom") + V4BAS(c) + V5BAS(c,"dom") + V6BAS(c) + V0MAR_CSI(c);

! Excerpt 13 of TABLO input file: !
!..............................!

! Income-Side Components of GDP !

Variable ! Aggregate Revenue, indirect taxes on ....    !
 V1TAX_CSI  # Intermediate #;
 V2TAX_CSI  # Investment   #;
 V3TAX_CS   # Households   #;
 V4TAX_C    # Export       #;
 V5TAX_CS   # Other Demand #;
 V0TAR_C    # Aggregate Tariff Revenue #;
 V0TAX_CSI  # Aggregate Indirect Tax Revenue #;

! Some of these equations also cover the start of Excerpt 30 !
Formula & Equation S_V1TAX_CSI
 V1TAX_CSI  = Sum(c,COM,Sum(s,SRC,Sum(i,IND,  V1TAX(c,s,i))));
Formula & Equation S_V2TAX_CSI
 V2TAX_CSI  = Sum(c,COM,Sum(s,SRC,Sum(i,IND,  V2TAX(c,s,i))));
```

```
Formula & Equation S_V3TAX_CS
 V3TAX_CS    = Sum(c,COM,Sum(s,SRC,              V3TAX(c,s)));
Formula & Equation S_V4TAX_C
 V4TAX_C     = Sum(c,COM,                        V4TAX(c));
Formula & Equation S_V5TAX_CS
 V5TAX_CS    = Sum(c,COM,Sum(s,SRC,              V5TAX(c,s)));
Formula & Equation S_V0TAR_C
 V0TAR_C     = Sum(c,COM,                        V0TAR(c));
Formula & Equation S_V0TAX_CSI
 V0TAX_CSI   = V1TAX_CSI + V2TAX_CSI + V3TAX_CS
                         + V4TAX_C + V5TAX_CS + V0TAR_C;

Variable ! All-Industry Factor Cost Aggregates !
  V1CAP_I    # total payments to capital #;
  V1LAB_IO   # total payments to labour #;
  V1LND_I    # total payments to land #;
  V1OCT_I    # total other cost ticket payments #;
  V1PRIM_I   # total primary factor payments#;
  V0GDPINC   # nominal gdp from income side #;

 Formula & Equation S_V1CAP_I
  V1CAP_I    = Sum(i,IND,V1CAP(i));
 Formula & Equation S_V1LAB_IO
  V1LAB_IO   = Sum(i,IND,Sum(o,OCC,V1LAB(i,o)));
 Formula & Equation S_V1LND_I
  V1LND_I    = Sum(i,IND,V1LND(i));
 Formula & Equation S_V1OCT_I
  V1OCT_I    = Sum(i,IND,V1OCT(i));
 Formula & Equation S_V1PRIM_I
  V1PRIM_I   = V1LAB_IO + V1CAP_I + V1LND_I;
 Formula & Equation S_V0GDPINC
  V0GDPINC   = V1PRIM_I + V1OCT_I + V0TAX_CSI;

! Excerpt 15 of TABLO input file: !
!................................!

! Occupational composition of labour demand !
!$  Problem: for each industry i, minimize labour cost    !
!$           Sum(o,OCC, P1LAB(i,o)*X1LAB(i,o) )           !
!$    such that  X1LAB_O(i) = CES( All,o,OCC: X1LAB(i,o) ) !

Coefficient (All,i,IND) SIGMA1LAB(i)
        # CES substitution between skill types #;
Read SIGMA1LAB From File MDATA Header "SLAB";

Equation (Linear) E_x1lab
        # Demand for labour by industry and skill group #
(All,i,IND)(All,o,OCC)
p_x1lab(i,o) = p_x1lab_o(i) -
                          SIGMA1LAB(i)*[p_p1lab(i,o) - p_p1lab_o(i)];

Equation (Linear) E_p1lab_o
        # Price to each industry of labour composite #
(All,i,IND)
 V1LAB_O(i)*p_p1lab_o(i) = Sum(o,OCC,V1LAB(i,o)*p_p1lab(i,o));

! Excerpt 16 of TABLO input file: !
!................................!

! Primary factor proportions !
!$  X1PRIM(i) =                 !
!$    CES( X1LAB_O(i)/A1LAB_O(i), X1CAP(i)/A1CAP(i), X1LND(i)/A1LND(i) ) !

Coefficient (All,i,IND) SIGMA1PRIM(i)
                  # CES substitution, primary factors #;
Read SIGMA1PRIM From File MDATA Header "P028";

Equation (Linear) E_x1lab_o # Industry demands for effective labour #
(All,i,IND)  p_x1lab_o(i) - p_a1lab_o(i) =
p_x1prim(i) - SIGMA1PRIM(i)*[p_p1lab_o(i) + p_a1lab_o(i) - p_p1prim(i)];
```

```
Equation (Linear) E_x1cap  # Industry demands for capital #
(All,i,IND)  p_x1cap(i) - p_a1cap(i) =
  p_x1prim(i) - SIGMA1PRIM(i)*[p_p1cap(i) + p_a1cap(i) - p_p1prim(i)];
Equation (Linear) E_p1lnd  # Industry demands for land #
(All,i,IND)  p_x1lnd(i) - p_a1lnd(i) =
  p_x1prim(i) - SIGMA1PRIM(i)*[p_p1lnd(i) + p_a1lnd(i) - p_p1prim(i)];
Equation (Linear) E_p1prim
          # Effective price term for factor demand equations #
(All,i,IND)
 V1PRIM(i)*p_p1prim(i) = V1LAB_O(i)*{p_p1lab_o(i) + p_a1lab_o(i)} +
 V1CAP(i)*{p_p1cap(i)+p_a1cap(i)}+V1LND(i)*{p_p1lnd(i) + p_a1lnd(i)};

! Excerpt 17 of TABLO input file: !
!..............................!
! Import/domestic composition of intermediate demands !
!$  X1_S(c,i) = CES( All,s,SRC: X1(c,s,i)/A1(c,s,i) ) !

Coefficient (All,c,COM) SIGMA1(c)
                # Armington elasticities: Intermediate #;
Read SIGMA1 From File MDATA Header "1ARM";

Equation (Linear) E_x1  # Source-Specific Commodity Demands #
(All,c,COM)(All,s,SRC)(All,i,IND)
p_x1(c,s,i)-p_a1(c,s,i) = p_x1_s(c,i)
  - SIGMA1(c)*{p_p1(c,s,i)+p_a1(c,s,i) - p_p1_s(c,i)};

Equation (Linear) E_p1_s # Effective Price of Commodity Composite #
(All,c,COM)(All,i,IND)
p_p1_s(c,i) = Sum(s,SRC, S1(c,s,i)*[p_p1(c,s,i) + p_a1(c,s,i)]);

! Excerpt 18 of TABLO input file: !
!..............................!
! Top nest of industry input demands !
!$  X1TOT(i) = MIN( All,c,COM: X1_S(c,i)/{A1_S(c,s,i)*A1TOT(i)},  !
!$                             X1PRIM(i)/{A1PRIM(i)*A1TOT(i)},    !
!$                             X1OCT(i)/{A1OCT(i)*A1TOT(i)} )   !

Equation (Linear) E_x1_s  # Demands for Commodity Composites #
(All,c,COM)(All,i,IND)
      p_x1_s(c,i) - {p_a1_s(c,i) + p_altot(i)} = p_x1tot(i);

Equation (Linear) E_x1prim  # Demands for primary factor composite #
(All,i,IND)  p_x1prim(i) - {p_a1prim(i) + p_altot(i)} = p_x1tot(i);

Equation (Linear) E_x1oct  # Demands for other cost tickets #
(All,i,IND)  p_x1oct(i) - {p_a1oct(i) + p_altot(i)} = p_x1tot(i);

Equation (Linear) E_p1tot  # Zero pure profits in production #
(All,i,IND)
V1TOT(i)*{p_p1tot(i)-p_altot(i)} =
 Sum(c,COM,  V1PUR_S(c,i)  *{p_p1_s(c,i) + p_a1_s(c,i)})
           +  V1PRIM(i)     *{p_p1prim(i) + p_a1prim(i)}
           +  V1OCT(i)      *{p_p1oct(i) + p_a1oct(i)};

! Excerpt 19 of TABLO input file: !
!..............................!
! Output mix !

Coefficient (All,i,IND) SIGMA1OUT(i)
                # CET transformation elasticities #;
Read SIGMA1OUT From File MDATA Header "SCET";

Equation (Linear) E_q1  # Supplies of commodities by industries #
(all,c,COM)(all,i,IND)
p_q1(c,i) = p_x1tot(i) + SIGMA1OUT(i)*(p_p0(c,"dom") - p_p1tot(i));

Variable (All,i,IND) MAKE_C(i) # all production by industry i #;
```

```
Formula & Equation S_MAKE_C # all production by industry i #
(All,i,IND) MAKE_C(i) = Sum(c,COM,MAKE(c,i));

Equation (Linear) E_xltot # Average price received by industries #
(All,i,IND) MAKE_C(i)*p_pltot(i) = Sum(c,COM,MAKE(c,i)*p_p0(c,"dom"));

! Excerpt 20 of TABLO input file: !
!.....................................!

! Investment demands !
!$  X2_S(c,i) = CES( All,s,SRC: X2(c,s,i)/A2(c,s,i) ) !

Coefficient (All,c,COM) SIGMA2(c)
                        # Armington elasticities: Investment #;
Read SIGMA2 From File MDATA Header "2ARM";

Equation (Linear) E_x2  # Source-Specific Commodity Demands #
(All,c,COM)(All,s,SRC)(All,i,IND)
p_x2(c,s,i)-p_a2(c,s,i) - p_x2_s(c,i)
    = - SIGMA2(c)*{p_p2(c,s,i)+p_a2(c,s,i) - p_p2_s(c,i)};

Equation (Linear) E_p2_s  # Effective Price of Commodity Composite #
(All,c,COM)(All,i,IND)
p_p2_s(c,i) = Sum(s,SRC, S2(c,s,i)*[p_p2(c,s,i)+p_a2(c,s,i)]);

! Investment top nest !
!$  X2TOT(i) = MIN( All,c,COM: X2_S(c,i)/{A2_S(c,s,i)*A2TOT(i)} ) !

Equation (Linear) E_x2_s  # Demands for Commodity Composites #
(All,c,COM)(All,i,IND)
    p_x2_s(c,i) - {p_a2_s(c,i) + p_a2tot(i)} = p_x2tot(i);

Equation (Linear) E_p2tot  # Zero pure profits in investment #
(All,i,IND) V2TOT(i)*(p_p2tot(i) - p_a2tot(i)) =
 Sum(c,COM,  V2PUR_S(c,i)  *{p_p2_s(c,i)+p_a2_s(c,i)});

! Excerpt 21 of TABLO input file: !
!.....................................!

! Import/domestic composition of household demands !

!$ X1_S(c,i) = CES( All,s,SRC: X1(c,s,i)/A1(c,s,i) ) !

Coefficient (All,c,COM) SIGMA3(c)
                        # Armington elasticities: Households #;
Read SIGMA3 From File MDATA Header "3ARM";

Equation (Linear) E_x3  # Source-Specific Commodity Demands #
(All,c,COM)(All,s,SRC)
p_x3(c,s)-p_a3(c,s) = p_x3_s(c) -
                      SIGMA3(c)*{ p_p3(c,s)+p_a3(c,s) - p_p3_s(c) };

Equation (Linear) E_p3_s  # Effective Price of Commodity Composite #
(All,c,COM) p_p3_s(c) = Sum(s,SRC, S3(c,s)*[p_p3(c,s)+p_a3(c,s)]);

! Excerpt 22 of TABLO input file: !
!.....................................!

! Data and formulae for coefficients used in household
                                                demand equations !

Coefficient FRISCH # the Frisch 'parameter' #;
Read  FRISCH From File MDATA Header "P021";
Update (Change) FRISCH = FRISCH*[p_w3lux - p_w3tot]/100.0;

Coefficient ALPHA # share of supernumerary in total expenditure #;
Formula (Always)  ALPHA = -1/FRISCH;

Coefficient (All,c,COM) S3LUX(c) # Marginal household budget shares #;
```

```
Read S3LUX From File MDATA Header "P044";
Update (All,c,COM) S3LUX(c) = p_a3lux(c);

Zerodivide off;
Zerodivide Default 0.0;
Coefficient (All,c,COM) S3_S(c)
                        # shares in total household expenditure #;
Formula (Always) (All,c,COM) S3_S(c) = V3PUR_S(c)/V3TOT;
Zerodivide Off;

Coefficient (All,c,COM) EPS(c) # Household expenditure elasticities #;
Zerodivide Default 1.0;
Formula (Always) (All,c,COM) EPS(c) = S3LUX(c)/S3_S(c);
!  marginal/average shares !
Zerodivide Off;
Zerodivide Default 0.0;

Coefficient (All,c,COM) B3LUX(c)
  #supernumerary expenditure commodity c/total expenditure commodity c#;
Formula (Always) (All,c,COM) B3LUX(c) = ALPHA*EPS(c);

! Excerpt 23 of TABLO input file: !
!..............................!

! Commodity composition of household demand !

Equation (Linear) E_x3sub
                 # Subsistence Demand for composite commodities #
 (All,c,COM) p_x3sub(c) = p_q + p_a3sub(c);

Equation (Linear) E_x3lux  # Luxury Demand for composite commodities #
(All,c,COM) p_x3lux(c) + p_p3_s(c) = p_w3lux + p_a3lux(c);

Equation (Linear) E_x3_s
                  # Total Household demand for composite commodities #
(All,c,COM) p_x3_s(c) = B3LUX(c)*p_x3lux(c) + [1-B3LUX(c)]*p_x3sub(c);

Equation (Linear) E_utility
# Change in utility disregarding taste change terms #
   p_utility + p_q = Sum(c,COM, S3LUX(c)*p_x3lux(c));

Equation (Linear) E_a3lux  # default setting for luxury taste shifter #
(All,c,COM) p_a3lux(c) = p_a3sub(c) - Sum(k,COM, S3LUX(k)*p_a3sub(k));

Equation (Linear) E_a3sub
                  # default setting for subsistence taste shifter #
(All,c,COM) p_a3sub(c) = p_a3_s(c) - Sum(k,COM, S3_S(k)*p_a3_s(k));

! Excerpt 24 of TABLO input file: !
!..............................!

! Export and other final demands !

Set
 TRADEXP      # Traditional Export Commodities #      (C1, C2, C4, C6);
 NTRADEXP     # Non-Traditional Export Commodities # (C3, C5, C7 - C23);
Subset
 TRADEXP      is subset of COM;
 NTRADEXP     is subset of COM;

Coefficient ! Export Aggregates !
 V4TRADEXP        # Total Traditional Export earnings #;
 V4NTRADEXP       # Total Non-Traditional Export earnings #;
Formula (Always)
 V4TRADEXP = Sum(c,TRADEXP,  V4PUR(c));
Formula (Always)
 V4NTRADEXP = Sum(c,NTRADEXP,  V4PUR(c));

Coefficient (All,c,COM) EXP_ELAST(c)
  # Export Demand Elasticities: Typical Value -20.0 #;
```

```
Read EXP_ELAST From File MDATA Header "P018";

Equation (Linear) E_x4_A  # Traditional export demand functions #
(All,c,TRADEXP)
     p_x4(c) - p_f4q(c) = EXP_ELAST(c)*[p_p4(c) - p_phi - p_f4p(c)];

Equation (Linear) E_x4_B  # Non-Traditional export demand functions #
(All,c,NTRADEXP) p_x4(c) = p_x4_ntrad;

Equation (Linear) E_p4_ntrad
                  # Average Price of Non-Traditional exports #
     V4NTRADEXP*p_p4_ntrad = Sum(c,NTRADEXP, V4PUR(c)*p_p4(c));

Coefficient EXP_ELAST_NT # Non-Traditional Export Demand Elasticity #;
Read EXP_ELAST_NT From File MDATA Header "EXNT";

Equation (Linear) E_x4_ntrad
                # Demand for Non-Traditional export aggregate #
    p_x4_ntrad     = EXP_ELAST_NT*[p_p4_ntrad - p_phi - p_f4_ntrad];

Equation (Linear) E_x5  # "Other" demands #
(All,c,COM)(All,s,SRC) p_x5(c,s) = p_f5(c,s) + p_f5tot;

Equation (Linear) E_f5tot # Overall "Other" demands shift #
     p_f5tot = p_x3tot + p_f5tot2;

! Excerpt 25 of TABLO input file: !
!..............................!

! Margin demands !

Zerodivide off;
Zerodivide Default 0.5;

Formula & Equation E_almar  # Margins to producers #
(All,c,COM)(All,s,SRC)(All,i,IND)(All,m,MAR)
   almar(c,s,i,m)  = x1mar(c,s,i,m) / x1(c,s,i);

Formula & Equation E_a2mar  # Margins to capital creators #
(All,c,COM)(All,s,SRC)(All,i,IND)(All,m,MAR)
   a2mar(c,s,i,m)  = x2mar(c,s,i,m) / x2(c,s,i);

Formula & Equation E_a3mar  # Margins to households #
(All,c,COM)(All,s,SRC)(All,m,MAR)
   a3mar(c,s,m)  = x3mar(c,s,m) / x3(c,s);

Formula & Equation E_a4mar  # Margins to exports #
(All,c,COM)(All,m,MAR)
   a4mar(c,m)  = x4mar(c,m) / x4(c);

Formula & Equation E_a5mar  # Margins to "Other" users #
(All,c,COM)(All,s,SRC)(All,m,MAR)
   a5mar(c,s,m)  = x5mar(c,s,m) / x5(c,s);

Zerodivide off;
! Excerpt 26 of TABLO input file: !
!..............................!

! The price system !

Zerodivide default 0.0;

Formula # purchasers prices - producers #
(All,c,COM)(All,s,SRC)(All,i,IND)
p1(c,s,i) = p0(c,s) * t1(c,s,i)
+ Sum(m,MAR,p0(m,"dom") * x1mar(c,s,i,m)) / x1(c,s,i);

Equation E_p1 # purchasers prices - producers #
(All,c,COM)(All,s,SRC)(All,i,IND)
 p1(c,s,i)*[tiny + x1(c,s,i)] = p0(c,s) * t1(c,s,i) * x1(c,s,i)
+ Sum(m,MAR,p0(m,"dom") * x1mar(c,s,i,m)) ;
```

```
Formula  # purchasers prices - capital creators #
(All,c,COM) (All,s,SRC) (All,i,IND)
 p2(c,s,i) = p0(c,s) * t2(c,s,i)
  + Sum(m,MAR,p0(m,"dom") * x2mar(c,s,i,m)) / x2(c,s,i);

Equation E_p2 # purchasers prices - capital creators #
(All,c,COM) (All,s,SRC) (All,i,IND)
 p2(c,s,i)*[tiny + x2(c,s,i)] = p0(c,s) * t2(c,s,i)*x2(c,s,i)
  + Sum(m,MAR,p0(m,"dom") * x2mar(c,s,i,m)) ;

Formula # purchasers prices - households #
(All,c,COM) (All,s,SRC)
 p3(c,s) = p0(c,s) * t3(c,s)
  + Sum(m,MAR,p0(m,"dom") * x3mar(c,s,m)) / x3(c,s);

 Equation E_p3 # purchasers prices - households #
(All,c,COM) (All,s,SRC)
 p3(c,s)*[tiny+x3(c,s)] = p0(c,s) * t3(c,s)*x3(c,s)
  + Sum(m,MAR,p0(m,"dom") * x3mar(c,s,m)) ;

Formula # Zero pure profits in Exporting #
(All,c,COM)
 p4(c) = p0(c,"dom") * t4(c)
  + Sum(m,MAR,p0(m,"dom") * x4mar(c,m)) / x4(c);

Equation E_p4 # Zero pure profits in Exporting #
(All,c,COM)
 p4(c)*[tiny + x4(c)] = p0(c,"dom") * t4(c)*x4(c)
  + Sum(m,MAR,p0(m,"dom") * x4mar(c,m)) ;
! note that we refer to Export taxes,not subsidies !

Formula # Zero pure profits in distribution of other#
(All,c,COM) (All,s,SRC)
 p5(c,s) = p0(c,s) * t5(c,s)
  + Sum(m,MAR,p0(m,"dom") * x5mar(c,s,m)) / x5(c,s);

Equation E_p5 # Zero pure profits in distribution of other#
(All,c,COM) (All,s,SRC)
 p5(c,s)*[tiny + x5(c,s)] = p0(c,s) * t5(c,s)*x5(c,s)
  + Sum(m,MAR,p0(m,"dom") * x5mar(c,s,m)) ;

Equation E_p0_A # Zero pure profits in importing #
(All,c,COM) p0(c,"imp") = pf0cif(c) * phi * t0imp(c);

Zerodivide off;

! Excerpt 28 of TABLO input file: !
!..............................!

! Tax rate equations !

Zerodivide default 0.0;

Equation (Linear)
E_t1 # power of tax on sales to intermediate #
   (All,c,COM) (All,s,SRC) (All,i,IND)
                  p_t1(c,s,i) = p_f0tax_s(c) + p_f1tax_csi;
(Linear) E_t2 # power of tax on sales to investment #
   (All,c,COM) (All,s,SRC) (All,i,IND)
                  p_t2(c,s,i) = p_f0tax_s(c) + p_f2tax_csi;
(Linear) E_t3 # power of tax on sales to households #
   (All,c,COM) (All,s,SRC)    p_t3(c,s)     = p_f0tax_s(c) + p_f3tax_cs;
(Linear) E_t4_A # power of tax on sales to traditional exports #
   (All,c,TRADEXP)             p_t4(c)       = p_f0tax_s(c) + p_f4tax_trad;
(Linear) E_t4_B # power of tax on sales to non-traditional exports #
   (All,c,NTRADEXP)           p_t4(c)       = p_f0tax_s(c) + p_f4tax_ntrad;
(Linear) E_t5 # power of tax on sales to other #
  (All,c,COM) (All,s,SRC)        p_t5(c,s)     = p_f0tax_s(c) + p_f5tax_cs;
```

```
! Excerpt 29 of TABLO input file: !
!..............................!

! Indirect tax revenue !

Formula & Equation E_w1tax_csi
# revenue from indirect taxes on flows to intermediate # w1tax_csi =
Sum(c,COM, Sum(s,SRC, Sum(i,IND,p0(c,s) * x1(c,s,i) * (t1(c,s,i)-1))));

Formula & Equation E_w2tax_csi
# revenue from indirect taxes on flows to investment # w2tax_csi =
Sum(c,COM, Sum(s,SRC, Sum(i,IND,p0(c,s) * x2(c,s,i) * (t2(c,s,i)-1))));

Formula & Equation E_w3tax_cs
# revenue from indirect taxes on flows to households # w3tax_cs =
Sum(c,COM, Sum(s,SRC,p0(c,s) * x3(c,s) * (t3(c,s)-1)));

Formula & Equation E_w4tax_c
# revenue from indirect taxes on exports # w4tax_c =
Sum(c,COM,p0(c,"dom") * x4(c) * (t4(c)-1));

Formula & Equation E_w5tax_cs
# revenue from indirect taxes on flows to "Other" # w5tax_cs =
Sum(c,COM, Sum(s,SRC,p0(c,s) * x5(c,s) * (t5(c,s)-1)));

Formula & Equation E_w0tar_c
# tariff revenue # w0tar_c =
Sum(c,COM,pf0cif(c) * phi * x0imp(c) * (t0imp(c)-1));

! Excerpt 30 is now contained in Excerpt 13 !
!..........................................!

! Factor incomes and GDP !

! Excerpt 31 of TABLO input file: !
!..............................!

! GDP expenditure aggregates !

Variable(change) (All,i,IND) R_T(i) # investment/capital ratio #;
Read R_T From File MDATA Header "YBYK"; ! numbers like 0.07 !

Formula & Equation S_x2tot # total investment #
(All,i,IND) x2tot(i) = R_T(i) * x1cap(i);

Formula # price of total investment #
(All,i,IND) p2tot(i) = V2TOT(i) / x2tot(i);

Equation
(Linear) E_x2tot_i  # total real investment #
  V2TOT_I*p_x2tot_i = Sum(i,IND,V2TOT(i)*p_x2tot(i));
(Linear) E_p2tot_i  # investment price index #
  V2TOT_I*p_p2tot_i = Sum(i,IND,V2TOT(i)*p_p2tot(i));
(Linear) E_w2tot_i  # total nominal investment #
  p_w2tot_i = p_x2tot_i + p_p2tot_i;

(Linear) E_x3tot    # real consumption #
  V3TOT*p_x3tot = Sum(c,COM, Sum(s,SRC,V3PUR(c,s)*p_x3(c,s) ));
(Linear) E_p3tot    # consumer price index #
  V3TOT*p_p3tot = Sum(c,COM,Sum(s,SRC,V3PUR(c,s)*p_p3(c,s)));
(Linear) E_w3tot    # household budget constraint #
  p_w3tot = p_x3tot + p_p3tot;

(Linear) E_x4tot    # export volume index #
  V4TOT*p_x4tot = Sum(c,COM,V4PUR(c)*p_x4(c));
(Linear) E_p4tot    # exports price index, $A #
  V4TOT*p_p4tot = Sum(c,COM,V4PUR(c)*p_p4(c));
(Linear) E_w4tot    # A$ Border value of exports #
  p_w4tot = p_x4tot + p_p4tot;
```

```
(Linear) E_x5tot     # aggregate real "Other" demands #
  V5TOT*p_x5tot = Sum(c,COM,Sum(s,SRC,V5PUR(c,s)*p_x5(c,s) ));
(Linear) E_p5tot     # 'other' demands price index #
  V5TOT*p_p5tot = Sum(c,COM,Sum(s,SRC,V5PUR(c,s)*p_p5(c,s)));
(Linear) E_w5tot     # aggregate nominal value of "Other" demands #
  p_w5tot = p_x5tot + p_p5tot;

(Linear) E_x6tot     # inventories volume index: base period dollars #
  V6TOT*p_x6tot = 100*Sum(c,COM,p0(c,"dom")*c_x6(c));
(Linear) E_p6tot     # inventories price index #
  V6TOT*p_p6tot = Sum(c,COM,V6BAS(c)*p_p0(c,"dom"));
(Linear) E_w6tot     # aggregate nominal value of inventories #
  p_w6tot = p_x6tot + p_p6tot;

(Linear) E_x0cif_c   # CIF Import volume index, CIF weights #
  V0CIF_C*p_x0cif_c = Sum(c,COM,V0CIF(c)*p_x0imp(c));
(Linear) E_p0cif_c   # A$ CIF imports price index #
  V0CIF_C*p_p0cif_c = Sum(c,COM,V0CIF(c)*(p_phi+p_pf0cif(c)));
(Linear) E_w0cif_c   # A$ CIF value of imports #
        p_w0cif_c = p_x0cif_c + p_p0cif_c;

(Linear) E_x0gdpexp  # real GDP,expenditure side #
  V0GDPEXP*p_x0gdpexp =
   V3TOT*p_x3tot + V2TOT_I*p_x2tot_i + V5TOT*p_x5tot
    + V6TOT*p_x6tot + V4TOT*p_x4tot - V0CIF_C*p_x0cif_c;
(Linear) E_p0gdpexp  # price index for GDP,expenditure side #
  V0GDPEXP*p_p0gdpexp =
   V3TOT*p_p3tot + V2TOT_I*p_p2tot_i + V5TOT*p_p5tot
    + V6TOT*p_p6tot + V4TOT*p_p4tot - V0CIF_C*p_p0cif_c;
(Linear) E_w0gdpexp  # nominal GDP from expenditure side #
  p_w0gdpexp = p_x0gdpexp + p_p0gdpexp;

Display V0GDPINC; ! Display for checking purposes !
Display V0GDPEXP; ! Two should always be the same !

! Excerpt 32 of TABLO input file: !
!..............................!

! Trade balance and other aggregates !

Formula & Equation E_delB      # (balance of trade) #
  B =( V4TOT - V0CIF_C)/V0GDPEXP ;

Equation
(Linear) E_x0imp_c   # import volume index, duty paid weights #
  V0IMP_C*p_x0imp_c = Sum(c,COM,V0IMP(c)*p_x0imp(c));
(Linear) E_p0imp_c   # duty paid imports price index #
  V0IMP_C*p_p0imp_c = Sum(c,COM,V0IMP(c)*p_p0(c,"imp"));
(Linear) E_w0imp_c   # value of imports (duty paid) #
  p_w0imp_c = p_x0imp_c + p_p0imp_c;

(Linear) E_x1cap_i   # aggregate usage of capital,rental weights #
  V1CAP_I*p_x1cap_i = Sum(i,IND,V1CAP(i)*p_x1cap(i));
(Linear) E_p1cap_i   # average capital rental #
  V1CAP_I*p_p1cap_i = Sum(i,IND,V1CAP(i)*p_p1cap(i));

Equation (Linear) E_employ # employment by industry #
(All,i,IND) V1LAB_O(i)*p_employ(i) = Sum(o,OCC,V1LAB(i,o)*p_x1lab(i,o));

(Linear) E_employ_i  # aggregate employment,wage bill weights #
  V1LAB_IO*p_employ_i= Sum(i,IND,V1LAB_O(i)*p_employ(i));

(Linear) E_x1prim_i  # aggregate output: value-added weights #
  V1PRIM_I*p_x1prim_i = Sum(i,IND,V1PRIM(i)*p_x1tot(i));

(Linear) E_p0toft    # terms of trade #
  p_p0toft = p_p4tot - p_p0cif_c;

(Linear) E_p0realdev  # real devaluation #
  p_p0realdev =  p_p0cif_c - p_p0gdpexp;
```

```
! Excerpt 33 of TABLO input file: !
!..............................!

! Rates of Return !

Variable (All,i,IND) QCOEF(i) # ratio, gross to net rate of return #;
Read QCOEF From File MDATA Header "P027";

Coefficient (parameter)(All,i,IND) DR(i)
# depreciation factor for investment #;

Formula & Equation S_DR # depreciation factor for investment #
(All,i,IND) DR(i) =  p1cap(i) / p2tot(i) * [ 1.0 - 1.0 / QCOEF(i) ];

Formula
p1cap_i = 1;
x1cap_i = V1CAP_I / p1cap_i;


Formula & Equation E_p1cap # definition of rates of return to capital #
(All,i,IND) r1cap(i) = p1cap(i) / [p2tot(i) * QCOEF(i)];

Coefficient(parameter) (all,i,IND) BETA_R(i);
Read BETA_R From File MDATA Header "BETR";

Formula r1cap_i = 1 ;

Formula (All,i,IND) f1ret(i) = r1cap(i)/r1cap_i/
                               ((x1cap(i)/x1cap_i)^BETA_R(i)) ;

Equation E_r1cap # capital growth rates related to rates of return #
(All,i,IND)
 r1cap(i)  = [x1cap(i) / x1cap_i]^BETA_R(i) * f1ret(i) * r1cap_i;

! Excerpt 34 of TABLO input file: !
!..............................!

! Indexing equations !

Equation (Linear) E_p1lab # flexible setting of money wages #
(All,i,IND)(All,o,OCC)
p_p1lab(i,o)= p_p3tot + p_f1lab_io + p_f1lab_o(i) +
                                    p_f1lab_i(o) + p_f1lab(i,o);

Equation (Linear) E_ploct # Indexing of prices of "Other Cost" tickets #
(All,i,IND) p_ploct(i) = p_p3tot + p_floct(i); !assumes full indexation!

! Excerpt 35 of TABLO input file: !
!..............................!

! Investment/capital accumulation equations !

Coefficient (parameter) (All,i,IND) DEP(i) # depreciation factors #;
Read DEP From File MDATA Header "DPRC";  ! numbers like 0.95 !

Coefficient (INTEGER) T # number of years covered by simulation #;
Read T From Terminal;   ! entered by user at runtime !

Set(intertemporal) YEARS MAXIMUM SIZE 100 (yr[1] - yr[T]);
Set(intertemporal) YEARS1 SIZE 1 (yr[1]);
Set(intertemporal) YEARS2 MAXIMUM SIZE 99 (yr[2] - yr[T]);

Subset YEARS1 is subset of YEARS;
Subset YEARS2 is subset of YEARS;

Coefficient (all,y,YEARS) ORD(y) # = y for y = 1 to T #;
Formula (Always)(All,y,YEARS1) ORD(y) = 1;
Formula (Always)(All,y,YEARS2) ORD(y) = ORD(y-1) + 1;
```

```
Coefficient (All,i,IND) Z(i) # K(T)/K(0)#;
Formula (All,i,IND) Z(i) = 1;
Update (All,i,IND) Z(i) = p_x1cap(i);

Coefficient (parameter) (All,i,IND) K_0(i) # K(0) #;
Formula (All,i,IND) K_0(i) = x1cap(i);

Coefficient (parameter) (All,i,IND) Y_0(i) # Y(0) #;
Formula (All,i,IND) Y_0(i) = x2tot(i);

Coefficient (All,i,IND) R_0(i) # Y(0)/K(0)#;
Formula (All,i,IND) R_0(i) = R_T(i);

Coefficient (parameter) (All,i,IND) DEP_T(i) # DEP to the power of T #;
Formula (All,i,IND) DEP_T(i) = DEP(i)^T;

Coefficient (parameter) (All,i,IND) N_term(i) # useful constant #;
Formula (All,i,IND) N_term(i) =
    Sum(y,YEARS, DEP(i)^(T -ORD(y)) );!note y takes values 1 to T!
Coefficient (parameter) (All,i,IND) M_term(i) # useful constant #;
Formula (All,i,IND) M_term(i) =
    Sum(y,YEARS,([ORD(y)-1]/T)*DEP(i)^(T -ORD(y)) );
Variable (All,i,IND) K_TERM(i) # Fudge coefficient  #;
Formula & Equation S_K_TERM (All,i,IND)
 K_TERM(i) = K_0(i) * [DEP_T(i)-1.0] + Y_0(i) * N_term(i);
Formula Fudge = 0;
(All,i,IND) f_accum(i) = 1;
Equation E_x2tot # investment/capital accumulation #
(All,i,IND) x1cap(i) = [K_TERM(i) * Fudge
   + (x2tot(i) - Y_0(i)) * M_term(i) + K_0(i)] * f_accum(i);
! Excerpt 36 of TABLO input file: !
!.....!
! Debt accumulation equations !
Formula
   p0cif_c = 1;
Formula & Equation E_BT # Real Trade Deficit #
   BT = (V0CIF_C - V4TOT) / p0cif_c;
Variable (change) DEBT_RATIO    # Debt/GDP ratio #;
Read DEBT_RATIO From File MDATA Header "DGDP";
Variable DEBT              # Real Foreign Debt #;
Formula & Equation S_DEBT # Real Foreign Debt #
    DEBT = DEBT_RATIO * V0GDPEXP / p0cif_c;
Coefficient (parameter) DEBT0          # Original Real Foreign Debt #;
Formula DEBT0 = DEBT;
Coefficient (parameter) R_WORLD
# World Interest Rate: number like 1.04 (4% rate) #;
Read R_WORLD From File MDATA Header "RWLD";
Coefficient (parameter) N_DEBT         # Useful Constant #;
Formula N_DEBT = Sum(y,YEARS, R_WORLD^(T - ORD(y)) );
Coefficient (parameter) M_DEBT         # Useful Constant #;
Formula
 M_DEBT = Sum(y,YEARS,([ORD(y)-1]/T)*R_WORLD^(T - ORD(y)) );
Coefficient (parameter) B0             # Original Real Trade Deficit #;
Formula B0 = BT;

Equation E_Debt          # foreign debt #
  Debt = [DEBT0 * (R_WORLD^T - 1) + B0 * N_DEBT] * Fudge +
                                                  M_DEBT * (BT - B0);
! end of file !
```